
図 1 ビジネスチャット企業導入率ビジネスチャット企業導入率

1.はじめに

コミュニケーションツールの代表的なものとしてメールがあ

る。使用しているサービスは企業により様々であるが、恐らく

ほとんどの企業が導入・活用しているであろう。

近年では新たなコミュニケーションツールとしてチャットを

導入する企業が増えている。きっかけは、テレワークの実施

が大きな要因であったかもしれない。

しかし近年、テレワークは実施する企業が徐々に減少してい

る。一方で企業のチャットサービスの導入率は【図１】のよう

に増加している。つまり、チャットサービスはテレワークの有

無に関わらず、コミュニケーションツールのひとつとして認

識され、活用されていることが分かる。

そこで本稿では、業務アプリケーションとメール・チャットサ

ービスとの連携方法について紹介する。連携によるメリット

は次章にて記載する。

業務アプリケーションと
メール・チャットサービスの連携

1.はじめに
2. コミュニケーションツールの特徴
3. Indyの利用
4. メール送信機能の実装
5. チャット送信機能の実装
6. おわりに

図 2 メール・チャットサービス連携前業務フローメール・チャットサービス連携前業務フロー

2.コミュニケーションツールの特徴

メール・チャットにはそれぞれ異なる特徴がある。チャット

はリアルタイムでメッセージの同期が可能である。簡単な

連絡やグループでの共有、即座に確認すべきことがある場

合に使用が適している。一方でメールは、重要な連絡や長

文でメッセージを送る場合、記録として残す必要がある場

合に適している。

上記の特徴により、社外への連絡はメールを使用し、社内

への連絡はチャットを利用するなど、状況によってコミュ

ニケーションツールを使い分けている企業も多いのでは

ないだろうか。

しかし、日常の業務において使用するツールが増えること

は、複数サービス（アプリ）を切り替えて使い分けなければ

いけないというデメリットも生まれる。

例えば、業務アプリケーションでデータを照会し、照会結

果を基にメール内容を入力するシーンでは、業務アプリケ

ーションの起動及びメールサービスの起動が必要となる。

さらに社内への連絡が追加で必要となれば、チャットサー

ビスの起動も必要となる。PCの限られた画面スペースの

中で複数サービスを立ち上げることは業務効率の低下に

も繋がりかねない。

本稿で紹介する業務アプリケーションとの連携を行うと、

メールやチャットサービスの立ち上げが不要となる。よっ

て、円滑に業務を進めることが可能となる。

3.Indyの利用

業務アプリケーションと各サービスの連携にはDelphiに

標準付属しているIndyを利用する。Indyとは、オープンソ

ースのネットワーク関連コンポーネントである。

本稿の例では業務アプリケーションで見積書を出力し、先

方にメール送付するシーンを想定する。また、添付する見積

書は圧縮＋パスワード付与を行い、メール送信後は社内に

チャットで送信完了を報告する業務フローとする【図２】。

Indyを利用し、業務アプリケーションとの連携を実現する

と【図３】のようなフローとなる。では、次章より具体的な実

装方法について紹介していく。

本サンプルでは、Delphi/400 11 Alexandriaを使用し、メ

ールサービスはOffice365（Outlook）、チャットサービス

はChatworkを使用する。

D
elphi/400
前
 坂
  誠
 二



図 4 業務アプリケーション（改修前）業務アプリケーション（改修前）

図 3 メール・チャットサービス連携後業務フローメール・チャットサービス連携後業務フロー

4.メール送信機能の実装

本稿では、業務アプリケーションの仕様を

以下と仮定する。

見積照会で作成済みの見積データを検索

し、チェックを付けたデータ毎に見積書を

発行可能とする。見積照会では、見積書の

送付先である先方担当者の名前を表示し

ており、メールアドレスなどの紐づく担当

者情報をマスタにて保管している【図４】。

図 5 メール送信用コンポーネントの配置メール送信用コンポーネントの配置

4-1.コンポーネントの貼り付け

まずは、メール送信処理を実装するために必要なコンポー

ネントの貼り付けを行う。メール送信機能を実装したい画

面に、TIdSMTPとTIdSSLIOHandlerSocketOpenSSL、

TIdMessageを配置し、各プロパティ設定を行う【図５】。

配置した各コンポーネントの役割は以下の通りである。

smtpMail：TIdSMTP

メールサーバーとの接続及び送信処理の実施

sslMail：TIdSSLIOHandlerSocketOpenSSL

TLS/SSLを使用した暗号化通信を可能にする

msgMail：TIdMessage

送信先アドレスやメッセージ内容の設定

各コンポーネントに対するプロパティ設定について詳しく

解説していく。

smtpMailでは、メールサーバーへの接続情報について設

定している。本稿ではoffice365を使用するため、Hostは

「smtp.office365.com」、Portは「587」にて設定する。設

定値については使用するメールサービスによって異なるた

め、各メールサービスのヘルプなどでご確認いただきたい。

また、使用するメールサービスによっては２段階認証の設

定をメールサービス側で行い、外部アプリケーションで使

用するためのパスワードを生成する必要がある。その際は、

生成パスワードをPasswordプロパティに設定しなければ

ならない。

sslMailでは、暗号化通信を行うための設定を行う。本稿で

はSSL/TLS（STARTTLS）を使用して通信を行うため

smtpMailと同様にsslMailにHostやPortの設定を行う。

SSLVersionsプロパティにはTIdSSLIOHandlerSocket

O p e n S S Lで指定可能な最新バージョンである

「sslTLSv1_2」を設定する。

smtpMailのIOHandlerプロパティには「sslMail」を指定

し、UseTLSプロパティには「utUseExplicitTLS」を指定

する。しかし、UseTLSプロパティは暗号化方式などによ

り、設定値が異なる為、各メールサービスに合わせて設定

いただきたい。

ソース 1

メール送信ロジック実装前

msgMailでは、送信するメッセージの文字タイプや文字コー

ドについて設定している。ContentTypeプロパティには主に

以 下 の よう な 形 式 を 指 定 で き る が 、本 稿 で は

「multipart/mixed」を指定する。

text/plain：文字テキストのみのメール形式

text/html：htmlと同様に文字に装飾が可能な形式

multipart/mixed：複数のメール形式が混在可能な形式

CharSetプロパティでは文字コードを指定する。Unicode形

式に対応するため「UTF-8」を設定する。ContentTransfer

Encodingプロパティではメールの変換方式について指定で

きる。本稿ではメールにファイル添付を行う想定であるため、

エンコード方式は「BASE64」を設定する。

4-2.メール送信ロジックの実装

コンポーネントの貼り付け完了後、次は送信ロジックの実装

を行う。実装は「見積書出力ボタン」押下時のタイミングで行

う。【ソース１】が送信ロジック実装前の処理である。本稿で

は、自動送信とするため、件名及び本文を予めAS/400にデ

ータ登録しておく【図６】。追加のロジックとしては、【ソース

２】【ソース３】の通り

①メール内容のデータを取得

②SMTPサーバーへの接続

③メールの送信元と送信先のセット

④メールの件名と本文のセット

⑤送信の実行

⑥SMTPサーバーの接続解除

である。また②の実施後は⑥を必ず実行する必要があるため

try～finally～endにて処理している。また、メール件名の文

字化けを防ぐために、msgMailのOnInitializeISOイベント

に処理を追加する【ソース４】。メール送信のロジックは以上

で完了である。しかし、実行すると【図７】のようなエラーが発

生する。

{*******************************************************************************

目的: 見積書発行ボタン押下時処理

引数:

戻値:

*******************************************************************************}

procedure TfrmReview.btnMTROutputClick(Sender: TObject);

begin

mtList.DisableControls;

mtList.First;

try

while not mtList.Eof do

begin

if mtList.FieldByName('CHEK').AsInteger = 1 then

begin

// 見積書出力(見積№のフォルダ)

OutPutMTMR;

end;

mtList.Next;

end;

finally

mtList.EnableControls;

end;

end;



図 6 登録メッセージ登録メッセージ

ソース 2
メール送信処理の実装（メイン処理）

{*******************************************************************************
目的: 見積書発行ボタン押下時処理
引数:
戻値:

*******************************************************************************}
procedure TfrmReview.btnMTROutputClick(Sender: TObject);
var

sSubject, sGetBody, sBody: String;
sConpany: String;
sTANTNM: String;
sMYNAME: String;

begin
// メール内容取得
sSubject := GetMSG('MTMRMAILK'); // 件名
sGetBody := GetMSG('MTMRMAILH'); // 本文

mtList.DisableControls;
mtList.First;
try

// SMTPサーバーへの接続
smtpMail.Connect; ②
try

while not mtList.Eof do
begin

if mtList.FieldByName('CHEK').AsInteger = 1 then
begin

// 見積書出力(見積№のフォルダ) 
OutPutMTMR;

GetMSG関数は
ソース3で記述

①



ソース 3
メール送信処理の実装（メール内容の取得）

try

// メールFrom - To
msgMail.From.Address := 'mgrtechnical@outlook.jp'; // メールFrom
msgMail.Recipients.Clear;                          // クリア
msgMail.Recipients.EMailAddresses :=
mtList.FieldByName('TTTNML').AsString;           // メールTo

msgMail.Subject := sSubject; // 件名セット

// メール本文
// 動的な値をStringReplaceで変換
sConpany := mtList.FieldByName('TKNM').AsString; // 得意先名
sTANTNM := mtList.FieldByName('STNM').AsString; // 先方担当者
sMYNAME := '前坂'; // 送信者名

msgMail.Body.Clear;
sBody := StringReplace(sGetBody, '[CANPANY]', sConpany, [rfReplaceAll]);
sBody := StringReplace(sBody, '[TANTNM]', sTANTNM, [rfReplaceAll]);
sBody := StringReplace(sBody, '[MYNAME]', sMYNAME, [rfReplaceAll]);
msgMail.Body.Text := sBody; // 本文セット

// 送信処理
smtpMail.Send(msgMail);

except
on E: Exception do
begin

raise;
end;

end;
end;

mtList.Next;
end;

finally
// SMTPサーバー接続解除
smtpMail.Disconnect;

end;
finally

mtList.EnableControls;
end;

end;

③

④

⑤

⑥

{*******************************************************************************

目的: メッセージマスタ取得処理

引数:

戻値:

*******************************************************************************}

function TfrmReview.GetMSG(AGetKEY: String): String;

begin

qryMSG.Close;

qryMSG.SQL.Text :=

' SELECT MSMESG FROM MMESGF WHERE MSSKEY = :MSSKEY ORDER BY MSMSRW ';

qryMSG.ParamByName('MSSKEY').AsString := AGetKEY;

qryMSG.Open;

try

while not qryMSG.Eof do 

begin



ソース 4

図 7 メール送信エラーメール送信エラー

メール件名の文字化け防止

if Result <> '' then

begin

// 改行

Result := Result + #13#10;

end;

Result := Result + qryMSG.FieldByName('MSMESG').AsString;

qryMSG.Next;

end;

finally

qryMSG.Close;

end;

end;

{*******************************************************************************
目的: TIdMessage Initialize処理
引数:
戻値:

*******************************************************************************}
procedure TfrmReview.msgMailInitializeISO(var VHeaderEncoding: Char;

var VCharSet: string);
begin

VHeaderEncoding := 'B';
// メールヘッダーのCharSetはここで設定する(メール件名に影響)
VCharSet := 'UTF-8';

end;



図 8 dllのインストールdllのインストール

図 9 メール送信実行結果メール送信実行結果

【 図 ７ 】の エ ラ ー を 解 決 す る た め に は

　　　　　　　　　　　　　より「libeay32 .d l l」

「ssleay32.dll」を取得し、実行モジュールと同階層に配置す

る必要がある【図８】。各dllを配置した後、再度モジュールを

実行するとメール送信がエラーなく実行される【図９】。https://indy.fulgan.com/SSL/



図 10 コンポーネントのインストールコンポーネントのインストール

図 11 ファイル圧縮用コンポーネントの貼り付けファイル圧縮用コンポーネントの貼り付け

4-3.出力フォルダの圧縮とパスワード付与

本稿では、出力した見積書は特定のフォルダに保管される

仕組みとする。メールには該当フォルダを圧縮後、パスワ

ードを付与したものを添付する。

フォルダの圧縮には「TAbZipper」コンポーネントを使用

する。「TAbZipper」は「GetItパッケージマネージャー」か

ら「Abbrevia 2021.11」を選択しインストールする。イン

ストール方法は【図１０】のとおりである。インストールが

完了したら、画面に「TAbZipper」コンポーネントを貼り付

けてみよう【図１１】。



ソース 5

ソース 6

ランダムパスワード作成処理

フォルダ圧縮処理

コンポーネントの貼り付けが完了したら、まずはランダムパ

スワードを作成する関数を実装する。ランダムな文字列を設

定するにはRandom関数を使用する。Random関数の呼び

出し前には、Randomize手続きにより乱数生成関数を初期

化しておく【ソース5】。ランダムパスワードの実装が完了した

ら、次は圧縮化の処理を実装する。圧縮化の処理は

FileNameに圧縮後のファイル名（フルパス）を指定し、

BaseDirectoryに圧縮対象のフォルダ（フルパス）を指定す

る。その後、AddFilesにて圧縮化するファイルを指定し、

SaveにてFileNameで指定した場所に保管される。本稿の

ようにPassWord設定を行いたい場合は、PassWordに文字

列をセットする【ソース6】。この時に併せて、設定パスワード

とZipファイル名を戻り値にしておく（メールへの添付処理

で使用）。

{*******************************************************************************

目的: ランダムパスワード作成処理

引数:

戻値: 生成パスワード

*******************************************************************************}

function TfrmReview.GetRandomPass: String;

var

sRandPassStr: String;

begin

Randomize;

sRandPassStr := 'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ1234567890';

repeat

Result := Result + sRandPassStr[Random(Length(sRandPassStr)) + 1];

until (Length(Result) = 10);

end;

{*******************************************************************************

目的: フォルダ圧縮処理

引数: AZipPassWord - 設定パスワード 戻り値

戻値: 圧縮ファイル名（フルパス）

*******************************************************************************}

function TfrmReview.ZipFileCreate(var AZipPassWord: String): String;

var

LFileName: TFileName;

sZipFileName: String;

begin

abzDir.FileNameȾ:= 'C:\Document\見積書\'Ⱦ+ mtList.FieldByName('MTNO').AsStringȾ+ '.zip';Ⱦ

abzDir.BaseDirectoryȾ:= 'C:\Document\見積書\'Ⱦ+ mtList.FieldByName('MTNO').AsString;



ソース 7
ファイル添付処理

4-4.圧縮ファイルの添付

圧縮ファイル作成の実装が完了したら、メールへの添付処

理 を 実 装 する 。メー ル へ の 添 付 は U s e s 節 に

IdAttachmentFileを追加し、TIdAttachmentFileを生

成する。後は、圧縮ファイル名及び各プロパティを設定す

るだけで実装完了である【ソース7①】。また、パスワードメ

ールも送付するよう併せて実装する【ソース7②】。ファイル

添付処理を実装した結果が【図１２】である。

// abzDir.BaseDirectory直下のファイルを全て圧縮

for LFileName in TDirectory.GetFiles(

abzDir.BaseDirectory, '*.*', TSearchOption.soAllDirectories) do

begin

sZipFileName :=

StringReplace(LFileName,

IncludeTrailingPathDelimiter(abzDir.BaseDirectory), '', [rfIgnoreCase]);

abzDir.AddFiles(sZipFileName, 0);

end;

abzDir.Password := GetRandomPass; // ランダムパスワード設定

AZipPassWord := abzDir.Password;

abzDir.Save;

Result := abzDir.FileName;

abzDir.CloseArchive;

end;

{*******************************************************************************

目的: 見積書発行ボタン押下時処理

引数:

戻値:

*******************************************************************************}

procedure TfrmReview.btnMTROutputClick(Sender: TObject);

var

sSubject, sGetBody, sBody: String;

sConpany: String;

sTANTNM: String;

sMYNAME: String;

sZipFileName, sZipPassWord: String;

begin

// メール内容取得

sSubject := GetMSG('MTMRMAILK'); // 件名

sGetBody := GetMSG('MTMRMAILH'); // 本文

mtList.DisableControls;

mtList.First;

try

// SMTPサーバーへの接続

smtpMail.Connect;

try

while not mtList.Eof do 

begin

if mtList.FieldByName('CHEK').AsInteger = 1 then 

begin

// 見積書出力(見積№＋システム日時のフォルダ) 

OutPutMTMR;

// Zip化（Zipファイル名を返却）

sZipFileName := ZipFileCreate(sZipPassWord); 

try

～ メール本文設定処理 省略 ～

// 添付ファイルの設定

msgMail.MessageParts.Clear;

with TIdAttachmentFile.Create(msgMail.MessageParts, sZipFileName) do 

begin
FileName := sZipFileName;

ContentType := 'application/octet-stream'; 

ContentTransfer := 'base64';

end;

// メール本文送信処理
smtpMail.Send(msgMail);

// パスワードメールの送付

msgMail.Subject := sSubject; // 件名セット

msgMail.Body.Clear;

msgMail.Body.Text := 'パスワードを送付致します。' + #13#10 + sZipPassWord; // 本文セット

msgMail.MessageParts.Clear;  // 添付ファイルクリア

// パスワードメール送信処理

smtpMail.Send(msgMail);

except

on E: Exception do 

begin

raise;

end;

end; 

end;

mtList.Next

end;;

finally

// SMTPサーバー接続解除

smtpMail.Disconnect;

end;

finally

mtList.EnableControls; 

end;

end;

①

②



図 13 チャット送信用コンポーネントの配置チャット送信用コンポーネントの配置

図 12 ファイル添付処理実行結果ファイル添付処理実行結果

MIGARO Technical Report   065

5.チャット送信機能の実装

本章では、チャット送信機能の実装方法について紹介して

いく。

5-1.コンポーネントの貼り付け

まずは、チャット送信処理を実装するために必要なコンポー

ネントの貼り付けを行う。TIdHTTPとTIdSSLIOHandler

SocketOpenSSLを配置し、各プロパティ設定を行う

【図１３】。配置した各コンポーネントの役割は以下の通り

である。

httpChat：TIdHTTP

Postメソッドを使用し、指定URLにデータ送信を実施

sslChat：TIdSSLIOHandlerSocketOpenSSL

TLS/SSLを使用した暗号化通信を可能にする

図 14 トークンの確認方法トークンの確認方法

図 15 ルームIDの確認方法ルームIDの確認方法

5-2.APIエンドポイントの確認

APIエンドポイントとは、Webサービスやアプリケーションに

外部クライアントが通信するためのURLである。つまり、本稿

で使用するChatworkにおいても、業務アプリケーションが

通信するためのURLを確認しなければならない。Chatwork

では、　　　　　　　　　　　　　　　　　　　　　　　

のページにて、設定するURLやパラメータに関する記載があ

る。また、チャット送信を行うためには併せてトークン、チャッ

トのルームIDを確認する必要がある。

トークンとはAPIの認証を提供するセキュリティの仕組みで

あり、APIを使用するためのパスワードのようなものである。

トークンの取得はサービスによって異なるが、Chatworkの

場合は【図１４】のように、サービス連携のリンクから確認す

ることで取得可能である。

ルームIDはチャットを行うグループの識別IDであり、ルーム

IDの指定により、指定グループへのチャット送信が可能とな

る、ルームIDの確認方法は【図１５】である。

https://developer.chatwork.com/docs/endpoints



ソース 8
チャット送信処理（呼び出し元）

5-3.チャット送信ロジックの実装
チャット送信ロジックの実装は先ほどのAPIエンドポイン

トを確認したページを参考にする。リクエストに必要な

ContentTypeプロパティとCustomHeadersプロパティ

に値を設定する。後は、送信先URLと送信するメッセージ

内容を引数にPostメソッドを実行すると、実装は完了であ

る【ソース８】【ソース９】。実行結果は【図１６】となる。

{*******************************************************************************
目的: 見積書発行ボタン押下時処理
引数:
戻値:

*******************************************************************************}
procedure TfrmReview.btnMTROutputClick(Sender: TObject);
var

sSubject, sGetBody, sBody: String;
sConpany: String;
sTANTNM: String;
sMYNAME: String;
sZipFileName, sZipPassWord: String;
sSendMTNO: String;

begin
// メール内容取得
sSubject := GetMSG('MTMRMAILK'); // 件名
sGetBody := GetMSG('MTMRMAILH'); // 本文

mtList.DisableControls;
mtList.First;
try

// SMTPサーバーへの接続
smtpMail.Connect;
try

while not mtList.Eof do
begin

if mtList.FieldByName('CHEK').AsInteger = 1 then
begin

// 見積書出力(見積№＋システム日時のフォルダ)
OutPutMTMR;

～ メール送信処理 省略 ～

// 送付見積№
if sSendMTNO <> '' then
begin

sSendMTNO := sSendMTNO + #13#10;
end;
sSendMTNO := sSendMTNO + mtList.FieldByName('MTNO').AsString

end;

mtList.Next;
end;

finally
// SMTPサーバー接続解除
smtpMail.Disconnect;

end;
finally

mtList.EnableControls;
end;

// チャット送信処理
SendChat(sSendMTNO);

end;

SendChat処理は
ソース9で記述



ソース 9

図 16 チャット送信処理実行結果チャット送信処理実行結果

チャット送信処理（メイン処理）

{*******************************************************************************
目的: チャット送信処理
引数: AMTNO : 送付見積№ ※複数可能性あり
戻値:

*******************************************************************************}
procedure TfrmReview.SendChat(AMTNO: String);
var

slBody: TStringList;
sSetInfo: String;
sURL, sTOKN, sRMID: String;

begin

sURL := 'https://api.chatwork.com/v2/rooms/';  // URL
 // トークン
 // ルームID

sTOKN := '7fe7a578371e92cdaa609e34b0ae43a2'; 
sRMID := '334476678'; 

slBody := TStringList.Create;
try

sSetInfo := 'body=[info]';
sSetInfo := sSetInfo + '[title]見積書の件[/title]';

sSetInfo := sSetInfo +‘見積№：’+ #13#10 + AMTNO + #13#10 + 'メール送信しました。';
sSetInfo := sSetInfo + '[/info]';
slBody.Add(sSetInfo);

httpChat.Request.ContentType := 'application/x-www-form-urlencoded';
httpChat.Request.CustomHeaders.AddValue('X-ChatWorkToken', sTOKN);
try

httpChat.Post(sURL + sRMID + '/messages', slBody);
except

raise;
end;

finally
FreeAndNil(slBody);

end;

end;

6.おわりに

各サービスのメッセージ送信画面を立ち上げることなく、

業務アプリケーション内で処理が完結することをおわかり

いただけただろうか。また、各サービスとの連携方法につ

いても、Indyを利用すると簡単に実装が可能であることを

実感いただけたと思う。

本稿は、送信の自動化による例でご紹介したが、もちろん

自身でメッセージ送信用の入力画面を作成いただくことも

可能である。また、コンポーネントを変更すれば受信機能

についても容易に実装可能である。

現在は、メールサービス、チャットサービスの種類が豊富

で、どれを導入すべきか判断が難しい。そこで、現在使用し

ている業務アプリケーションと連携可能かといった観点を

ひとつの判断基準としてもよいのかもしれない。

本稿でご紹介した実装方法であれば、プロパティの設定

値を変更するだけで他のサービスへの連携も可能である。

もし、現在ご検討されているメールサービスやチャット

サービスがあれば、本稿を参考に連携可能かをお試しい

ただけると幸いである。

